/* This example is PURELY I LLUSTRATIVE: it’s purpose is to introduce as much of the syntax and semantics of ACL as possible, NOT to make the simplest or most correct contracts. The notion of the correctness of a contract is not well-defined and needs to be explored further. As is the issue of how abstract (vs programmatic) ACL really is…
*/

Import Core;

Namespace My.Examples

{ 

/*An abstract contract is NOT bound to a type of the IUT. Also, T will be bound upon ContainerBased being refined.  */
abstract Contract IndexableContainerBase<Type T>

{ 

  Scalar Integer size; 
//number of elements in container
// ACL supports scalars (something with a value) and lists.
/*An observability is a query-method that is used to provide state information about the IUT. That is, they are read-only methods that acquire and return a value stored by the IUT. It’s a bridge from the abstract TRM to a specific implementation.   */

  Observability Boolean  
IsFull();

  Observability Boolean  
IsEmpty();

  Observability T
ItemAt(Integer index); 

//the previous observability forces indexability but not a specific implementation
  Observability Integer  
Size();

 abstract Observability Boolean  HasItem(T aItem); 

//an abstract observability MUST be refined in a derived contract
/* the body of the "new" responsibility is executed immediately following the creation of a new contract instance.

*/

 Responsibility new()

{ size = 0;   Post(IsEmpty() == true)

}
/* the body of the "finalize" responsibility is executed immediately before the destruction of the current contract  instance.  

Here we insist that items be deleted before the container is deallocated.  */
Responsibility finalize()

{ Pre(IsEmpty() == true); 
}

/* Invariants provide a way to specify a set of checks that are to be executed before and after the execution of all bound responsibilities. Invariants precede pre-conditions, and follow post-conditions. We follow EIFFEL semantics.


The keyword context refers the container itself: there is a contract space and an execution space that coexist as the IUT executes and is monitored. context allows to refer to variables of a contract.*/

Invariant SizeCheck

{ Check(context.size >= 0);

   Check(context.size == Size())

}

/* This responsibility defines pre- and post- conditions for any addition. It is not to be bound but rather to be extended by actual responsibilities.

The keyword ‘Execute indicates where execution occurs. */

Responsibility GenericAddition(T aItem)

{ Pre(aItem not= null);    Pre(IsFull() == false);  Execute();

   size = size + 1;

  Post(HasItem(aItem)); 

}

/*This responsibility extends GenericAddition. It therefore reuses the pre- and post-conditions of GenericAddition. 

It does not add any other checks to those of GenericAddition. But Add can (and will) be refined in the contract that extends the current abstract one.
*/

Responsibility Add(T aItem) extends GenericAddition(aItem)

{ Execute(); }

/* Insert also extends GenericAddition and thus reuses its pre- and post-conditions. 

But it also adds pre- and post-conditions of its own due to the fact that its interface involves the use of an index.   */

Responsibility Insert(Integer index, T aItem) 

extends GenericAddition(aItem)

{ Pre(index >= 0); 
Execute();

   Post(ItemAt(index) == aItem);

}

/* Responsibility Remove returns the element removed.

The keyword value denotes this return value. It’s a bridge at execution time between the contract world and the actual execution. */
Responsibility T Remove()

{ Pre(IsEmpty() == false); Execute();

   size = size - 1;

   Post(value not= null);

   Post(HasItem(value) == false);

} //assumes only 1 occurrence of value

Responsibility RemoveElement(T aItem)

{ Pre(IsEmpty() == false); Pre(HasItem(aItem) == true);

  Execute();

  size = size - 1;

  Post(HasItem(aItem) == false);

} //assumes only 1 occurrence of aItem
/* The following scenario merely consists of a trigger statement and a terminate statement. There is no grammar of responsibilities between these two statements (in contrast to most scenarios.)

This scenario captures that the addition of an element x must eventually be followed by removal of x.   There will be a scenario instance for each x.
Here Add or Insert trigger the scenario, and Remove or RemoveElement terminate it.

Notice the use of the ‘dontcare’ keyword for the first parameter of Insert.  */

Scenario AddAndRemove

{ once Scalar T x;

   Trigger(Add(x) | Insert(dontcare, x)),

   Terminate((x == Remove()) | (RemoveElement(x)));  }

}    

}  //end of contract IndexableContainerBase

/* A TRM must include a main contract. It typically includes several other contracts.

The main contract of a TRM must be bound to a type of the IUT . 
Here IndexableContainer inherits from IndexableContainerBase.

Single and multiple inheritance are supported for composing contracts together in ACL.  

Also, note that T in ContainerBase is explicitly  bound here to the type tItem (using syntax similar to templates in C++) */
MainContract IndexableContainer extends IndexableContainerBase<tItem>

{ List Integer container_times;

// keep a list of the amount of time that each item spends in the container.

   Scalar Timer item_timer;

/* Timer is a built-in type of our VF

A single timer can be used to time multiple items concurrently. */
   Scalar Integer number_of_items;

//used to store the total number of items that are stored by the container during execution 
// The abstract responsibility of ContainerBase is now refined: very programmatic
refine Observability Boolean HasItem(tItem item) //this refinement is NOT required!
{ tItem x;  Boolean result = false;

   loop(0 to Size())

{ x = ItemAt(counter);


   result = result || x == item;}
//we’re assuming unique valued items… see Adds

   value = result;
//we repeat, value is the keyword for return value
}

/* A parameter can be set explicitly, or using the binding tool, or be set at run-time. Here, it controls whether the static check below is to be performed or not. */


Parameters 

{ Scalar Boolean CheckMembers;
}

/* What follows is a static check that uses the built-in check HasMemberOfType to verify if the container holds instances of type tItem. This check is performed only if parameter CheckMembers is true. A belief is merely a message logged in the report (CER) produced by the VF. */

Structure

{ choice(Parameters.CheckMembers) == true

{ Belief CheckMember("There should be a member in our container to hold elements

 of type tItem")

{ HasMemberOfType(tItem); }    }    }  //this check needs to be investigated!!
/* We refine new: Pre- and post-conditions of the parent are checked before these ones. */

refine Responsibility new()

{  number_of_items = 0;

    container_times.Init();
 // create a list}

/* The ‘fire’ keyword is used to create an instance of an event that can, in turn, trigger or be observed in scenarios.  See below. */ 


refine Responsibility finalize()

{  fire(ContainerDone); 

}

/* Next, Add, Insert, Remove and RemoveElement from ContainerBase are further refined to use timers. 

More specifically, the scenario AddAndRemove (in the parent contract) creates an instance of itself for each element that is added to the container. This allows us to start a timer in Add or Insert upon insertion of an element and to stop that timer when that element is removed. In turn, this allows us to store the time spent by an element in the container.   */
refine Responsibility Add(tItem item) 

{ Pre(HasItem(item) == false); 
Execute();

   item_timer.Start(item); //built-in way to start a timer (identified by item)
  number_of_items = number_of_items + 1;
}

refine Responsibility Insert(Integer index, tItem item)

{ Pre(HasItem(item) == false); 
Execute();

   item_timer.Start(item);

  number_of_items = number_of_items + 1;   }

/* notice at THIS level, Add and Insert are the same: 

How could we avoid the following redundancy, if at all?  */

/* The keyword value below refers to the value returned by Remove()
refine Responsibility tItem Remove()

{ Execute();

  item_timer.Stop(value);  //the timer is identified by the value it is associated with
  container_times.Add(item_timer.Value(value));  
} //Value is value of timer…
refine Responsibility RemoveElement(tItem item)

{ Execute();

   item_timer.Stop(item);

  container_times.Add(item_timer.Value(item));  }

/* This responsibility is to be used in the scenario ContainerLifetime below. RemoveScn abstracts away which of the two Remove responsibilities is used. Contrast to a stub...
Notice again the use of keyword ‘dontcare’.    */


Responsibility RemoveScn()

{ Remove() | RemoveElement(dontcare);  }  //explicit list of alternatives
/* A stub responsibility is a place holder for one or more responsibilities. Here, we have only one choice, the default one, which is responsibility Add. Parameters and other mechanisms could be used to select between different kinds of addition, as illustrated elsewhere. */

stub Responsibility AddElement(tItem item)

{ Pre(item not= null);

  [Default] Add(item);

}

/* This scenario illustrates a Trigger being followed by a grammar of responsibilities and then a Terminate statement. In this case, the Terminate MUST be preceded by an ‘observe’ statement specifying the event that enables this termination. 

In the following scenario, a new scenario instance is created each time a new container is constructed (via the new responsibility). The responsibility new acts as the trigger.

The ‘,’ denotes the ‘follow’ operator. 

An atomic block defines a grammar of responsibilities so that no other responsibilities of this contract instance are allowed to execute except the ones specified within the grammar. 

The scenario must observe the event ContainerDone before concluding by proceeding with the execution of  finalize (which fires the event ContainerDone before its checks. This semantic contortion is due to the way scenario instances are monitored. 
There will be one instance of this scenario for each container created.*/

Scenario ContainerLifetime

{ Trigger(new()),

  atomic

    { (Add(dontcare) | Insert(dontcare, dontcare))*,

(RemoveScn())*; },


//is this semantically correct??
  observe(ContainerDone),

  Terminate(finalize());

} 

/* the metric returns THE list of integers representing the amount of time that each element spent in our container.
*/

Metric List Integer ContainerTimes()

{ context.container_times;}

// the metric returns THE total number of items that were stored in the container.

Metric Scalar Integer NumberOfItems()

{ context.number_of_items;

}

//This section of contract is to build the evaluation report.


Reports

// {0} is where the reported result goes in the output string

{ Report(
"The average time in the container is {0} milliseconds", 
AvgMetric(ContainerTimes())); //built-in AvgMetric

/* A report all statement performs the exact same way as the report statement, except that it generates a single result for all contract instances.   */

   ReportAll("The average time in all containers is {0} milliseconds", 
AvgMetric(ContainerTimes()));

  Report("The number of items added to the container is {0}", 
NumberOfItems());

  ReportAll("The number of items added to all containers is {0}", 
NumberOfItems());


}

/* The type tItem used for the elements of the container cannot be type bound to the container nor any of its descendants. So, here, we do not allow lists of lists. */

Exports

{ Type tItem conforms Item

{ not context;  not derived context;
} 
}



} }

In the container example to download we also find an inter-scenario interaction consisting of a single relation: for better or worse it hardwires that 8 instances of the AddAndRemove scenario and 4 instances of the ContainerLifetime scenario must execute independently. */

Import Core;

Namespace DaveArnold.Examples

{

    Interaction ContainerInter

    {



Relation MultipleContainers



{ // || is the independence or concurrency operator




Contract Container c;




c.AddAndRemove[8] || c.ContainerLifetime[4];

}  }

